

Syntax & semantics of Beginning Student

Readings: HtDP, Intermezzo 1 (Section 8).

We are covering the ideas of section 8, but not the parts of it dealing with section 6/7 material (which will come later), and in a somewhat different fashion.

Topics:

- Modelling programming languages
- Racket's semantic model
- Substitution rules (so far)

Modelling programming languages

- A program has a precise meaning and effect.
- A model of a programming language provides a way of describing the meaning of a program.
- Typically this is done informally, by examples.
- With Racket, we can do better.

Advantages in modelling Racket

- Few language constructs, so model description is short
- We don't need anything more than the language itself!
 - No diagrams
 - No vague descriptions of the underlying machine

Spelling rules for Beginning Student

Identifiers are the names of constants, parameters, and user-defined functions.

They are made up of letters, numbers, hyphens, underscores, and a few other punctuation marks. They must contain at least one non-number. They can't contain spaces or any of these:

`() , ; { } [] ' ' " "`.

Symbols start with a single quote ' followed by something obeying the rules for identifiers.

There are rules for numbers (integers, rationals, decimals) which are fairly intuitive.

There are some built-in constants, like `true` and `false`.

Of more interest to us are the rules describing program structure.

For example: a program is a sequence of definitions and expressions.

Syntax, semantics, and ambiguity

There are three problems we need to address:

1. Syntax: The way we're allowed to say things.
'?is This Sentence Syntactically Correct'
2. Semantics: the meaning of what we say.
'Trombones fly hungrily.'
3. Ambiguity: valid sentences have exactly one meaning.
'Sally was given a book by Joyce.'

English rules on these issues are pretty lax. For Racket, we need rules that *always* avoid these problems.

Grammars

To enforce syntax and avoid ambiguity, we can use *grammars*.

For example, an English sentence can be made up of a subject, verb, and object, in that order.

We might express this as follows:

$$\langle \text{sentence} \rangle = \langle \text{subject} \rangle \langle \text{verb} \rangle \langle \text{object} \rangle$$

The linguist Noam Chomsky formalized grammars in this fashion in the 1950's. The idea proved useful for programming languages.

The textbook describes function definitions like this:

$$\langle \text{def} \rangle = (\text{define } (\langle \text{var} \rangle \langle \text{var} \rangle \dots \langle \text{var} \rangle) \langle \text{exp} \rangle)$$

There is a similar rule for defining constants. Additional rules define **cond** expressions, etc.

The Help Desk presents the same idea as

$$\text{definition} = (\text{define } (\text{id id id } \dots) \text{ expr})$$

In CS 135, we will use informal descriptions instead.

CS 241, CS 230, CS 360, and CS 444 discuss the mathematical formalization of grammars and their role in the interpretation of computer programs and other structured texts.

Racket's semantic model

The second of our three problems (syntax, semantics, ambiguity) we will solve rigorously with a semantic model. A semantic model of a programming language provides a method of predicting the result of running any program.

Our model will repeatedly simplify the program via substitution.

Every substitution step yields a valid program (in full Racket), until all that remains is a sequence of definitions and values.

A substitution step finds the leftmost subexpression eligible for rewriting, and rewrites it by the rules we are about to describe.

Application of built-in functions

We reuse the rules for the arithmetic expressions we are familiar with to substitute the appropriate value for expressions like $(+ 3 5)$ and $(\text{expt } 2 10)$.

$(+ 3 5) \Rightarrow 8$

$(\text{expt } 2 10) \Rightarrow 1024$

Formally, the substitution rule is:

$(f v_1 \dots v_n) \Rightarrow v$ where f is a built-in function and v is the value of $f(v_1, \dots, v_n)$.

Note the two uses of an *ellipsis* $(. . .)$. What does it mean?

Ellipses

For built-in functions f with one parameter, the rule is:

$(f v_1) \Rightarrow v$ where v is the value of $f(v_1)$

For built-in functions f with two parameters, the rule is:

$(f v_1 v_2) \Rightarrow v$ where v is the value of $f(v_1, v_2)$

For built-in functions f with three parameters, the rule is:

$(f v_1 v_2 v_3) \Rightarrow v$ where v is the value of $f(v_1, v_2, v_3)$

We can't just keep writing down rules forever, so we use ellipses to show a *pattern*:

$(f v_1 \dots v_n) \Rightarrow v$ where v is the value of $f(v_1, \dots, v_n)$.

Application of user-defined functions

As an example, consider $(\text{define (term } x y) (* x (\text{sqr } y)))$.

The function application $(\text{term } 2 3)$ can be evaluated by taking the body of the function definition and replacing x by 2 and y by 3.

The result is $(* 2 (\text{sqr } 3))$.

The rule does not apply if an argument is not a value, as in the case of the second argument in $(\text{term } 2 (+ 1 2))$.

Any argument which is not a value must first be simplified to a value using the rules for expressions.

The general substitution rule is:

$(f\ v_1 \dots v_n) \Rightarrow \text{exp}'$

where $(\text{define } (f\ x_1 \dots x_n)\ \text{exp})$ occurs to the left, and exp' is obtained by substituting into the expression exp , with all occurrences of the formal parameter x_i replaced by the value v_i (for i from 1 to n).

Note we are using a pattern ellipsis in the rules for both built-in and user-defined functions to indicate several arguments.

Example:

```
(define (term x y) (* x (sqr y)))  
(term (- 3 1) (+ 1 2))  
⇒ (term 2 (+ 1 2))  
⇒ (term 2 3)  
⇒ (* 2 (sqr 3))  
⇒ (* 2 9)  
⇒ 18
```

A constant definition binds a name (the constant) to a value (the value of the expression).

We add the substitution rule:

$\text{id} \Rightarrow \text{val}$, where (define id val) occurs to the left.

Example:

```
(define x 3) (define y (+ x 1)) y
⇒
(define x 3) (define y (+ 3 1)) y
⇒
(define x 3) (define y 4) y
⇒
(define x 3) (define y 4) 4
```

Substitution in cond expressions

There are three rules: when the first expression is false, when it is true, and when it is **else**.

$(\text{cond} [\text{false exp}] \dots) \Rightarrow (\text{cond} \dots).$

$(\text{cond} [\text{true exp}] \dots) \Rightarrow \text{exp}.$

$(\text{cond} [\text{else exp}]) \Rightarrow \text{exp}.$

These suffice to simplify any **cond** expression.

Here the ellipses are serving a different role. They are not showing a pattern, but showing an *omission*. The rule just says “whatever else appeared after the **[false exp]**, you just copy it over.”

Example:

```
(define n 5) (define x 6) (define y 7)

(cond [(even? n) x][(odd? n) y])
⇒ (cond [(even? 5) x][(odd? n) y])
⇒ (cond [false x][(odd? n) y])
⇒ (cond [(odd? n) y])
⇒ (cond [(odd? 5) y])
⇒ (cond [true y])
⇒ y
⇒ 7
```

Example:

```
(define n 5) (define x 6)

(cond [(even? n) x][(odd? n) y])
⇒ (cond [(even? 5) x] [(odd? n) y])
⇒ (cond [false x][(odd? n) y])
⇒ (cond [(odd? n) y])
⇒ (cond [(odd? 5) y])
⇒ (cond [true y])
⇒ y
⇒ y: this variable is not defined
```

Errors

A syntax error occurs when a sentence cannot be interpreted using the grammar.

A run-time error occurs when an expression cannot be reduced to a value by application of our (still incomplete) evaluation rules.

Example:

```
(cond [(> 3 4) x])
⇒ (cond [false x])
⇒ (cond )
⇒ cond: all question results were false
```

Simplification rules for **and** and **or**

The simplification rules we use for Boolean expressions involving **and** and **or** are different from the ones the Stepper in DrRacket uses.

The end result is the same, but the intermediate steps are different.

The implementers of the Stepper made choices which result in more complicated rules, but whose intermediate steps appear to help students in lab situations.

$(\text{and false } \dots) \Rightarrow \text{false}.$

$(\text{and true } \dots) \Rightarrow (\text{and } \dots).$

$(\text{and}) \Rightarrow \text{true}.$

$(\text{or true } \dots) \Rightarrow \text{true}.$

$(\text{or false } \dots) \Rightarrow (\text{or } \dots).$

$(\text{or}) \Rightarrow \text{false}.$

As in the rewriting rules for **cond**, we are using an omission ellipsis.

Substitution rules (so far)

1. Apply functions only when all arguments are values.
2. When given a choice, evaluate the leftmost expression first.
3. $(f v_1 \dots v_n) \Rightarrow v$ when f is built-in...
4. $(f v_1 \dots v_n) \Rightarrow \text{exp}'$ when $(\text{define } (f x_1 \dots x_n) \text{exp})$ occurs to the left...
5. $\text{id} \Rightarrow \text{val}$ when (define id val) occurs to the left.

6. $(\text{cond } [\text{false exp}] \dots) \Rightarrow (\text{cond } \dots).$

7. $(\text{cond } [\text{true exp}] \dots) \Rightarrow \text{exp}.$

8. $(\text{cond } [\text{else exp}]) \Rightarrow \text{exp}.$

9. $(\text{and false } \dots) \Rightarrow \text{false}.$

10. $(\text{and true } \dots) \Rightarrow (\text{and } \dots).$

11. $(\text{and}) \Rightarrow \text{true}.$

12. $(\text{or true } \dots) \Rightarrow \text{true}.$

13. $(\text{or false } \dots) \Rightarrow (\text{or } \dots).$

14. $(\text{or}) \Rightarrow \text{false}.$

Importance of the model

We will add to the semantic model when we introduce a new feature of Racket.

Understanding the semantic model is very important in understanding the meaning of a Racket program.

Doing a step-by-step reduction according to these rules is called **tracing** a program.

It is an important skill in any programming language or computational system.

We will test this skill on assignments and exams.

Goals of this module

You should understand the substitution-based semantic model of Racket, and be prepared for future extensions.

You should be able to trace the series of simplifying transformations of a Racket program.